Expérimentations Mathematicales

Journées « La science informatique pour tous les lycéens »

7 avril 2014 - Yves Papegay - INRIA Sophia Antipolis

Congrès Maths en jean

Le pelage de la girafe (ou les diagrammes de Voronoï)

un peu d'aide



Percolation



un peu d'aide



Quelques caractéristiques

```
4 + 7
```

```
11
```

```
Plot[Sin[x], {x, -Pi, Pi}]
```




```
Plot3D[Sin[x] Cos[y], {x, -Pi, Pi}, {y, -Pi, Pi}]
```



Arithmétiques

entière

```
500 !
```

```
1 220 136 825 991 110 068 701 238 785 423 046 926 253 574 342 803 192 842 192 413 588 385 845 373 ⋱
  153 881 997 605 496 447 502 203 281 863 013 616 477 148 203 584 163 378 722 078 177 200 480 785 ⋱
  205 159 329 285 477 907 571 939 330 603 772 960 859 086 270 429 174 547 882 424 912 726 344 305 ⋱
  670 173 270 769 461 062 802 310 452 644 218 878 789 465 754 777 149 863 494 367 781 037 644 274 ⋱
  033 827 365 397 471 386 477 878 495 438 489 595 537 537 990 423 241 061 271 326 984 327 745 715 ⋱
  546 309 977 202 781 014 561 081 188 373 709 531 016 356 324 432 987 029 563 896 628 911 658 974 ⋱
  769 572 087 926 928 871 281 780 070 265 174 507 768 410 719 624 390 394 322 536 422 605 234 945 ⋱
  850 129 918 571 501 248 706 961 568 141 625 359 056 693 423 813 008 856 249 246 891 564 126 775 ⋱
  654 481 886 506 593 847 951 775 360 894 005 745 238 940 335 798 476 363 944 905 313 062 323 749 ⋱
  066 445 048 824 665 075 946 735 862 074 637 925 184 200 459 369 692 981 022 263 971 952 597 190 ⋱
  945 217 823 331 756 934 581 508 552 332 820 762 820 023 402 626 907 898 342 451 712 006 207 714 ⋱
  640 979 456 116 127 629 145 951 237 229 913 340 169 552 363 850 942 885 592 018 727 433 795 173 ⋱
  014 586 357 570 828 355 780 158 735 432 768 888 680 120 399 882 384 702 151 467 605 445 407 663 ⋱
  535 984 174 430 480 128 938 313 896 881 639 487 469 658 817 504 506 926 365 338 175 055 478 128 ⋱
  640 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 ⋱
  000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000 000
```

rationnelle (et plus)

```
3 / 4 + 2 / 3
```

$$\frac{17}{12}$$

```
Cos[Pi / 4]
```

$$\frac{1}{\sqrt{2}}$$

```
(3 + 5 I) / (2 - 3 I)
```

$$-\frac{9}{13}+\frac{19\,i}{13}$$

flottante

```
Cos[Pi / 4] // N
```

```
0.707107
```

Disgression 1 : Nombres flottants

Les nombres flottants sont caractérisés par deux entiers qui définissent la **base** et la **précision** de leur représentation.

Ainsi, en base 10 et avec une précision de 6, π est représenté par 3.14159.

```
Pi // N
```

```
3.14159
```

La plupart des machines représentent les nombres en base 2 - avec 52 bits pour la mantisse, 1 bit pour le signe et 11 bits pour l'exposant - ce qui a pour conséquence directe que seule une partie des nombres décimaux est représentable.

exemple de 0.1

Par exemple, en base 2, le nombre décimal 0.1 est représenté par

```
BaseForm[0.1`16, 2]
```

$0.00011001100110011001100110011001100110011001100110011010_2$

or, si l'on calcule exactement ce que cette représentation vaut exactement

```
RealDigits[0.00011001100110011001100110011001100110011001100110011010]
ToNumber[%, 2]
N[%, 60]
```

```
{{1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0,
  1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 0, 1, 1, 0, 1, 0}, -3}
```

```
3 602 879 701 896 397 / 36 028 797 018 963 968
```

```
0.100000000000000005551115123125782702118158340454101562500000
```

ce qui diffère légèrement de 0.1.

plus petit nombres

Cette représentation est par essence discrète et sa granularité dépends du nombre total de bits utilisés pour représenter un nombre (32 ou 64) et de la répartition des bits entre le signe, la mantisse et l'exposant.

```
$MachineEpsilon
```

2.22045×10^{-16}

```
N[2^-52]
```

2.22045×10^{-16}

Elle est habituellement définie comme le plus petit nombre qui ajouté à 1 produit un résultat strictement supérieur à 1.

```
x = 1. + $MachineEpsilon
y = 1. + ($MachineEpsilon / 2)
```

```
1.

1.
```

```
x - 1.
y - 1.
```

2.22045×10^{-16}

```
0.
```

Elle n'est pas absolue, mais relative au nombre au voisinage duquel on la calcule, aussi peut-on représenter des nombres plus petit que cette granularité et notamment, elle ne doit pas être confondue avec le plus petit nombre représentable.

```
$MinMachineNumber
RealDigits[%, 2]
```

2.22507×10^{-308}

```
{{1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
  0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, -1021}
```

Intervalles et arrondis

De fait, cette répartition discrète des nombres représentables en machine le long de la droite des réels conduit à associer à chaque nombre un intervalle, et à systématiquement arrondir tout réel au nombre représentable le plus proche.



Et l'implémentation de l'arithmétique et des fonctions numériques usuelles doit (ou devrait !) tenir compte de cette association, tant au niveau des arrondis de calculs que de la signification des résultats - ce qui n'est pas en général le cas.

arrondis

Voici deux exemples où l'arithmétique est prise en flagrant délit de ne pas considérer 1 comme le petit intervalle de la droite réelle qu'il représente.

```
Clear[x, y]
```

```
x = 3 (1 + $MachineEpsilon / 2)
```

```
3.
```

```
x = 3 + 3 / 2 $MachineEpsilon
```

```
3.
```

```
x - 3
```

```
4.44089 × 10^-16
```

```
x - 3
```

```
0.
```

```
x = (1 + $MachineEpsilon / 2) ^ 2
```

```
1.
```

```
x - 1
```

```
0.
```

signification

Dans cette exemple, la largeur de l'intervalle associé à 2.22×10^{25}est de l'ordre de 10^9, ce qui enlève tout sens au calcul du sinus.

```
Sin[2.22 × 10 ^ 25]
Sin[2.22 × 10 ^ 25 + Pi / 2]
Sin[2.22 × 10 ^ 25 + Pi]
```

```
0.34318
```

```
0.34318
```

```
0.34318
```

Arithmétique par intervalles

L'arithmétique par intervalle est mathématiquement bien définie comme une extension ensembliste de l'arithmétique sur les réels.
Ainsi, par exemple, la somme de deux intervalles est le plus petit (au sens de l'inclusion) intervalle qui contient la somme de deux nombres quelconques de ces deux intervalles.

Une définition similaire permet d'étendre les foncions numériques usuelles aux intervalles.

réels et intervalles

On peut associer chaque réel au plus petit intervalle dont les bornes sont des nombres représentables en machine qui le contient.

```
Interval[1.] // InputForm
```

```
Interval[{0.9999999999999999, 1.0000000000000002}]
```

Et si l'implémentation de l'arithmétique par intervalle est correcte - c'est à dire prends en compte les arrondis, et c'est en général le cas - alors on obtient des résultats qui retrouvent tout leur sens et sont intrinsèquement justes.

```
3 Interval[1.] - 3
```

```
Interval[{-8.88178 × 10^-16, 1.33227 × 10^-15}]
```

```
Sin[Interval[2.22 × 10^25]]
```

```
Interval[{-1, 1}]
```

Un calcul pour frémir

L'instabilité numérique de certains calculs est parfois intrinsèque aux expressions que l'on évalue, parfois relié à la manière dont on conduit cette évaluation, mais rarement prévisible simplement ou évidente. En voici un exemple qui illustre bien la nécessité de certifier les résultats des calculs.

Le polynome ci-dessous est dû a Rump.

```
RumpFunc[x_, y_] :=
  (1335 / 4 - x^2) y^6 + x^2 (11 x^2 y^2 - 121 y^4 - 2) + (11 / 2) y^8 + x / (2 y)
RumpFunc[77 617, 33 096]
% // N
```

$$-\frac{54767}{66192}$$

```
-0.827396
```

Ce premier calcul est exacte puisqu'il est réalisé avec des nombre entiers et rationnels. Voici le même calcul avec des nombres flottants.

```
RumpFunc[77 617., 33 096.]
```

```
0.
```

Plus troublant encore est le résultat obtenu si l'on remplace 11/2 par 5.5 dans la définition du polynome.

```
RumpFuncN[x_, y_] :=
  (1335 / 4 - x^2) y^6 + x^2 (11 x^2 y^2 - 121 y^4 - 2) + (5.5) y^8 + x / (2 y)
RumpFuncN[77 617, 33 096]
```

1.18059×10^{21}

Le recours a une arithmétique d'intervalles permet d'obtenir un résultat juste et de comprendre le phénomène ... même si il n'est pas très intéressant d'un point de vue calculatoire.

```
RumpFunc[Interval[77 617.], Interval[33 096.]]
```

$\text{Interval}\left[\left\{-3.89595 \times 10^{22}, 3.65983 \times 10^{22}\right\}\right]$

Quelques caractéristiques

Arithmétiques

entière

rationnelle (et plus)

flottante

flottante en précision controlée

```
RumpFunc[77617.`20, 33096.`20]
```

$0. \times 10^{18}$

```
RumpFunc[77617.`30, 33096.`30]
```

$0. \times 10^{8}$

```
RumpFunc[77617.`40, 33096.`40]
```

-0.8

```
RumpFunc[77617.`50, 33096.`50]
```

-0.82739605995

symbolique

```
Clear[x, y]
```

```
2 x + 1 - x
```

1 + x

```
expr = (x^2 + 6 x + 9) / (x^2 - 9)
```

$$\frac{9 + 6\, x + x^2}{-9 + x^2}$$

```
Simplify[expr]
```

$$\frac{3 + x}{-3 + x}$$

```
Factor[expr]
```

$$\frac{3 + x}{-3 + x}$$

```
expr2 = (x - 1) Sum[x^i, {i, 1, 15}]
```

$(-1+x)\,(x+x^2+x^3+x^4+x^5+x^6+x^7+x^8+x^9+x^{10}+x^{11}+x^{12}+x^{13}+x^{14}+x^{15})$

```
Simplify[expr2]
```

$x\,(-1+x^{15})$

```
Expand[expr2]
```

$-x+x^{16}$

un langage simple

des types simples

- entiers
- rationnels
- complexes
- flottants
- symboles

```
toto, α, x12, ℂ
```

- chaines de caracteres

```
"Bonjour, ça va ?"
```

une seule règle grammaticale

```
expr[arg1, arg2, ..., argN]
```

un peu de douceur syntaxique (optionelle)

```
Plus[a, b]
a + b
```

```
a + b
```

```
a + b
```

```
List[a, b, c]
{a, b, c}
```

```
{a, b, c}
```

```
{a, b, c}
```

```
x = 4
Set[y, 5]
```

```
4
```

```
5
```

```
x
y
```

```
4
```

```
5
```

```
Clear[x, y]
```

```
f[x]
f@x
x // f
```

```
f[x]
```

```
f[x]
```

```
f[x]
```

```
ReplaceAll[x + 3 x y + y^2, Rule[y, Sin[x]]]
x + 3 x y + y^2 /. y → Sin[x]
```

$x + 3\, x\, \mathrm{Sin}[x] + \mathrm{Sin}[x]^2$

$x + 3\, x\, \mathrm{Sin}[x] + \mathrm{Sin}[x]^2$

```
Function[x, x^2][4]
(#^2 &)[4]
```

```
16
```

```
16
```

une visualisation arborescente

```
Expand[1 + (x + Sin[y])^2] // FullForm
```

```
Plus[1, Power[x, 2], Times[2, x, Sin[y]], Power[Sin[y], 2]]
```

```
Expand[1 + (x + Sin[y])^2] // TreeForm
```



Un langage universel et homogène

graphique

```
inref[pt1_, pt2_, k_] := {Re[#], Im[#]} &[#2 ((#2 / #1) ^k) &[pt1.{1, I}, pt2.{1, I}]]
N[Chop[{-1 / 2, 1 / 3} / (1 - Unitize[#] + #)] &[Norm[{-1 / 2, 1 / 3}]]]
```

```
{-0.83205, 0.5547}
```

```
With[{inrefs =
    inref[{-0.8320502943378437`, 0.5547001962252291`}, {1, 0}, #] & /@ Range[-1, 88]},
 Graphics[Flatten@{GrayLevel[.6], Disk[], White, Line[inrefs]},
  PlotRange → 3 / 2, Background → GrayLevel[.3]]]
```



```
Graphics[
 Table[{Hue[RandomReal[]], Circle[RandomReal[4, {2}], RandomReal[1]]}, {40}]]
```



```
Graphics[{EdgeForm[Black], Table[{Hue[RandomReal[]], Rotate[
     Scale[Rectangle[RandomReal[2, 2]], RandomReal[]], RandomReal[2 Pi]]}, {20}]}]
```



```
Graphics3D[Table[With[{p = {i, j, k} / 5},
   {RGBColor[p], Opacity[.75], Cuboid[p, p + .15]}], {i, 5}, {j, 5}, {k, 5}]]
```



```
ContourPlot3D[x^4 + y^4 + z^4 - (x^2 + y^2 + z^2)^2 + 3 (x^2 + y^2 + z^2) == 3,
 {x, -2, 2}, {y, -2, 2}, {z, -2, 2}, Mesh → None,
 ContourStyle → Directive[Orange, Opacity[0.8], Specularity[White, 30]]]
```



images

img =  ;

```
Plot3D[1 / (x^2 + y^2 + .05), {x, -1, 1}, {y, -1, 1}, PlotStyle → Texture[img],
 PlotRange → All, Mesh → False, Boxed → False, Axes → False, Lighting → "Neutral"]
```

sons

```
PieChart[Table[With[{j = i}, Button[1, EmitSound[Sound@SoundNote[j]]]], {i, 14}]]
```



Speak[]

graphes

```
SocialMediaData["Facebook", "FriendNetwork"]
```



documents

Un peu de texte

```
nb = CreateDocument[];
```

```
NotebookWrite[nb, Cell["Bonjour", "Subsection"], All]
```

```
CreateDocument[
 Cases[NotebookGet[InputNotebook[]], Cell[___, "Section" | "Subsection"], Infinity]]
```

```
NotebookObject[ Untitled-9 ]
```

Un peu de dynamisme

variables dynamiques

```
Dynamic[x]
```

x

```
Slider[Dynamic[x]]
```



```
Dynamic@Module[{hour, min, sec, ht, mt, st}, Clock[];
  {hour, min, sec} = Take[DateList[], -3]; ht = Pi / 2 - 2 Pi hour / 12 - 2 Pi min / 720;
  mt = Pi / 2 - 2 Pi min / 60; st = Pi / 2 - 2 Pi Floor[sec] / 60; Graphics[{Arrowheads[0.1],
    Arrow[{{0, 0}, 0.6 {Cos[ht], Sin[ht]}}], Arrow[{{0, 0}, 0.85 {Cos[mt], Sin[mt]}}],
    Line[{{0, 0}, 0.85 {Cos[st], Sin[st]}}], PointSize[Medium],
    Table[Point[0.9 {Cos[i], Sin[i]}], {i, 0, 2 Pi, Pi / 6}], Point[{0, 0}], Circle[]}]]
```



objets interactifs

```
Slider[]
```

```
DynamicModule[{x = ""},
 PopupMenu[Dynamic[x], CountryData["Countries"], "Choose one…"]]
```



L' effet Manipulate

```
Manipulate[Plot[f[3 x + Pi / 2], {x, -Pi, Pi}], {f, {Sin, Cos}}]
```




```
Manipulate[Plot[f[a x + ϕ], {x, -Pi, Pi}], {ϕ, 0, 2 Pi}, {a, 1, 5}, {f, {Sin, Cos, Tan}}]
```



```
Manipulate[Plot3D[1 / ((x - x0) ^2 + (y - y0) ^2 + .05), {x, -1, 1},
  {y, -1, 1}, PlotStyle → Texture[img], PlotRange → All, Mesh → False,
  Boxed → False, Axes → False, Lighting → "Neutral"], {x0, -1, 1}, {y0, -1, 1}]
```



```
Manipulate[
 Dynamic[p = N[Chop[p / (1 - Unitize[#] + #)] &[Norm[p]]];
  q = N[Chop[q / (1 - Unitize[#] + #)] &[Norm[q]]];
  With[{inrefs = inref[p, q, #] & /@ Range[-1, s]},
   Graphics[Flatten@{GrayLevel[.6], Disk[], White, Line[inrefs]}, PlotRange → 3 / 2,
    Background → GrayLevel[.3]]]], {{p, {-1 / 2, 1 / 3}, "first"}, Locator},
 {{q, {1, 0}, "second"}, Locator},
 {{s, 1, "step"}, 0, 120, 1, Appearance → "Labeled"},
 AutorunSequencing → {3}, SaveDefinitions → True]
```



Retour sur Math en Jean’s

Girafe

```
Manipulate[ListDensityPlot[Map[Flatten, Transpose[{loc, Range[Length[loc]]}]],
  PlotRange → {{0, 10}, {0, 10}}, InterpolationOrder → 0, Mesh → All,
  ColorFunction → (ColorData["BeachColors"][#] &), FrameTicks → False],
 {{loc, RandomReal[{0, 10}, {3, 2}]}, {0, 0}, {10, 10},
  Locator, LocatorAutoCreate → True}]
```



Percolation

```
Grid[Table[Table[
   With[{mat = Map[If[# < 41, 1, 0] &, RandomInteger[{1, 100}, {50, 50}], {2}]}, With[
     {mat2 = MorphologicalComponents[mat /. {1 → 0, 0 → 1}, CornerNeighbors → False]},
     ArrayPlot[mat2, PixelConstrained → True, ColorRules →
       Join[Thread[DeleteCases[Union[mat2[[1]]], 0] → Blue], {0 → Black, _ → White}],
      ColorFunction → "Rainbow"]]], {4}], {3}]]
```



J'aurai aussi aimé vous parler de ....

Where I am ?

WolframAlpha::timeout :
The call to WolframAlpha[Where I am ?] has exceeded 30.` seconds. Increasing the value of the TimeConstraint option may improve the result. ≫

```
$Failed
```

```
$ImportFormats
$ExportFormats
```

```
demonstrations.wolfram.com
```

Merci de votre attention.